// File: /contracts/ScumbugsNFT.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "erc721a/contracts/ERC721A.sol";
import "./Pausable.sol";
import "./ScumbugsMetadata.sol";
import "./OpenseaProxy.sol";

/**
* @notice Smart contract for Scumbugs.
* ╭━━━┳━━━┳╮╱╭┳━╮╭━┳━━╮╭╮╱╭┳━━━┳━━━╮
* ┃╭━╮┃╭━╮┃┃╱┃┃┃╰╯┃┃╭╮┃┃┃╱┃┃╭━╮┃╭━╮┃
* ┃╰━━┫┃╱╰┫┃╱┃┃╭╮╭╮┃╰╯╰┫┃╱┃┃┃╱╰┫╰━━╮
* ╰━━╮┃┃╱╭┫┃╱┃┃┃┃┃┃┃╭━╮┃┃╱┃┃┃╭━╋━━╮┃
* ┃╰━╯┃╰━╯┃╰━╯┃┃┃┃┃┃╰━╯┃╰━╯┃╰┻━┃╰━╯┃
* ╰━━━┻━━━┻━━━┻╯╰╯╰┻━━━┻━━━┻━━━┻━━━╯
*/
contract ScumbugsNFT is ERC721A, Pausable, ScumbugsMetadata {

    // ---------------------- Variables ----------------------
    address private proxyRegistryAddress;
    uint256 public unitPrice = 0.069 ether;
    uint32 public constant supply = 15341;
    uint16 public constant maxAmount = 20;
    bool private isOpenSeaProxyActive = true;

    
    // ---------------------- Constructor ----------------------

    /**
     * @notice Constructor.
     */
    constructor(bytes32 _siteUrl, address _ScumbugsValues, address _proxyRegistryAddress) 
    ERC721A("Scumbugs", "SCUMBUGS")
    ScumbugsMetadata(_siteUrl, _ScumbugsValues)
    {
        proxyRegistryAddress = _proxyRegistryAddress;
        _safeMint(msg.sender, 341);
    }

    /**
     * @notice Returns the tokenId of the last minted Scumbug.
     */
    function currentTokenId() view external returns (uint256) {
        return _currentIndex - 1;
    }

    /**
     * @notice Mints new NFT(s).
     */
    function mintNFT(uint16 amount) external payable whenNotPaused {
        require(amount <= maxAmount, "Can't mint more than max amount");
        require(msg.value >= (unitPrice * amount), "Value should be equal or greater than unit price * amount");
        require((_currentIndex + amount - 1) < supply, "Can't mint that amount of NFTs");
        _safeMint(msg.sender, amount);
    }

    /**
     * @notice Withdraws the funds in the contract and sends them to the contract's owner
     */
    function withdraw() external {
        payable(owner()).transfer(address(this).balance);
    }

    /**
     * @notice Generates and stores metadata for given token id
     * @param metadataInput The data of the generated Scumbug
     */
    function generateMetadata_w1o(MetadataInput memory metadataInput) external onlyOwner {
        uint256 tokenId = metadataInput.tokenId;
        bool generated = isGenerated(tokenId);
        require(!generated, "Metadata already generated for tokenId");
        require(_currentIndex > tokenId, "NFT with tokenId does not exist");
        delete generated;
        attributesMap[tokenId] = Attributes({
            hand: metadataInput.hand,
            body: metadataInput.body,
            eyes: metadataInput.eyes,
            head: metadataInput.head,
            mouth: metadataInput.mouth,
            background_color: metadataInput.background_color,
            bug_type: metadataInput.bug_type,
            birthday: metadataInput.birthday
        });
        txhashes[tokenId] = metadataInput.txHash;
        mediaIds1[tokenId] = metadataInput.mediaId1;
        mediaIds2[tokenId] = metadataInput.mediaId2;
        mediaBdayIds1[tokenId] = metadataInput.mediaBdayId1;
        mediaBdayIds2[tokenId] = metadataInput.mediaBdayId2;
    }

    /**
     * @notice Sets siteUrl new value
     */
    function setSiteUrl(bytes32 _siteUrl) external onlyOwner {
        siteUrl = _siteUrl;
    }

    /**
     * @notice Sets the unit price
     */
    function setUnitPrice(uint256 _unitPrice) external onlyOwner whenPaused {
        unitPrice = _unitPrice;
    }

    /**
     * @dev Override tokenURI to return json instead of URI.
     * Taken from GenesisRepot: https://etherscan.io/address/0x5d4683ba64ee6283bb7fdb8a91252f6aab32a110#code
     */
    function tokenURI(uint256 tokenId) public view override returns (string memory) {
        require(_exists(tokenId), 'ERC721Metadata: URI query for nonexistent token');
        return _render(tokenId);
    }

    // function to disable gasless listings for security in case
    // opensea ever shuts down or is compromised
    // from CryptoCoven https://etherscan.io/address/0x5180db8f5c931aae63c74266b211f580155ecac8#code
    function setIsOpenSeaProxyActive(bool _isOpenSeaProxyActive) public onlyOwner {
        isOpenSeaProxyActive = _isOpenSeaProxyActive;
    }

    /**
     * @dev Override isApprovedForAll to allowlist user's OpenSea proxy accounts to enable gas-less listings.
     * Taken from CryptoCoven: https://etherscan.io/address/0x5180db8f5c931aae63c74266b211f580155ecac8#code
     */
    function isApprovedForAll(address owner, address operator)
        public
        view
        override
        returns (bool)
    {
        // Get a reference to OpenSea's proxy registry contract by instantiating
        // the contract using the already existing address.
        ProxyRegistry proxyRegistry = ProxyRegistry(proxyRegistryAddress);

        if (isOpenSeaProxyActive && address(proxyRegistry.proxies(owner)) == operator) {
            return true;
        }

        return super.isApprovedForAll(owner, operator);
    }
 
}
// File: /contracts/ScumbugsValues.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

/**
 * @title Scumbugs Data Values
 */
contract ScumbugsValues {

    mapping(uint8 => bytes32) public handMap;
    mapping(uint8 => bytes32) public bodyMap;
    mapping(uint8 => bytes32) public eyesMap;
    mapping(uint8 => bytes32) public headMap;
    mapping(uint8 => bytes32) public mouthMap;
    mapping(uint8 => bytes32) public backgroundMap;
    mapping(uint8 => bytes32) public bugTypeMap;


    constructor() {
        // Initialize hand map
        handMap[0] = bytes32("No traits");
        handMap[1] = bytes32("Bug You");
        handMap[2] = bytes32("Smoke");
        handMap[3] = bytes32("Money Talk");
        handMap[4] = bytes32("Bug Spray");
        handMap[5] = bytes32("Camcorder");
        handMap[6] = bytes32("Golf Club");
        handMap[7] = bytes32("Spicey");
        handMap[8] = bytes32("Genius");
        handMap[9] = bytes32("All In");
        handMap[10] = bytes32("Fishing Rod");
        handMap[11] = bytes32("Lactose Intolerant");
        handMap[12] = bytes32("Pizza");
        handMap[13] = bytes32("Scum Drink");
        handMap[14] = bytes32("Selfie Stick");
        handMap[15] = bytes32("Fake Watch");
        handMap[16] = bytes32("Rings");
        handMap[17] = bytes32("Boujee");
        handMap[18] = bytes32("Scissorhands");
        handMap[19] = bytes32("Artist");
        handMap[20] = bytes32("Graffiti");
        handMap[21] = bytes32("Lasso");
        handMap[22] = bytes32("Mic");
        handMap[23] = bytes32("Wood");
        handMap[24] = bytes32("Spy");
        handMap[25] = bytes32("Lettuce Finger");
        handMap[26] = bytes32("Baller");
        handMap[27] = bytes32("Katana");
        handMap[28] = bytes32("Scary Puppet");
        handMap[29] = bytes32("Sippin");
        handMap[30] = bytes32("Shroom");
        handMap[31] = bytes32("Banana Gun");
        handMap[32] = bytes32("Flask");
        handMap[33] = bytes32("Boombox");
        handMap[34] = bytes32("Larry SB");
        handMap[35] = bytes32("Wonderful gm");
        handMap[36] = bytes32("Slingshot");
        handMap[37] = bytes32("Skateboard");
        handMap[38] = bytes32("Glow Stick");
        handMap[39] = bytes32("Sadomaso");
        handMap[40] = bytes32("Demi God");
        handMap[41] = bytes32("I Can Point This");

        // Initialize body map
        bodyMap[0] = bytes32("No traits");
        bodyMap[1] = bytes32("Logo T");
        bodyMap[2] = bytes32("System B");
        bodyMap[3] = bytes32("Daddy's Rich");
        bodyMap[4] = bytes32("McScumin");
        bodyMap[5] = bytes32("Stunt Bug");
        bodyMap[6] = bytes32("Bug Squad");
        bodyMap[7] = bytes32("Track Jacket");
        bodyMap[8] = bytes32("Polo");
        bodyMap[9] = bytes32("Boomer");
        bodyMap[10] = bytes32("Turtleneck");
        bodyMap[11] = bytes32("Coconut Bra");
        bodyMap[12] = bytes32("Tie-dye");
        bodyMap[13] = bytes32("Life Jacket");
        bodyMap[14] = bytes32("Down Jacket");
        bodyMap[15] = bytes32("Perfecto");
        bodyMap[16] = bytes32("Thunder Fleece");
        bodyMap[17] = bytes32("Spandex");
        bodyMap[18] = bytes32("Bugs Racing");
        bodyMap[19] = bytes32("Scum Bag");
        bodyMap[20] = bytes32("Trailer Park Bugs");
        bodyMap[21] = bytes32("Disco Shirt");
        bodyMap[22] = bytes32("Knit");
        bodyMap[23] = bytes32("Gloves Jacket");
        bodyMap[24] = bytes32("Sherpa Fleece");
        bodyMap[25] = bytes32("Velvet Hoodie");
        bodyMap[26] = bytes32("Hoodie Up");
        bodyMap[27] = bytes32("Denim Jacket");
        bodyMap[28] = bytes32("Biker Vest");
        bodyMap[29] = bytes32("Sherling Jacket");
        bodyMap[30] = bytes32("V God");
        bodyMap[31] = bytes32("Trench Coat");
        bodyMap[32] = bytes32("Tuxedo");
        bodyMap[33] = bytes32("Dragon Shirt");
        bodyMap[34] = bytes32("Scammer");
        bodyMap[35] = bytes32("B.W.A.");
        bodyMap[36] = bytes32("Punk Jacket");
        bodyMap[37] = bytes32("Loose Knit");
        bodyMap[38] = bytes32("Red Puffer");
        bodyMap[39] = bytes32("Wolf");
        bodyMap[40] = bytes32("King Robe");
        bodyMap[41] = bytes32("Notorious B.U.G.");
        bodyMap[42] = bytes32("Scum God Jacket");
        bodyMap[43] = bytes32("Fast Lane");
        bodyMap[44] = bytes32("Iced Out Chain");
        bodyMap[45] = bytes32("Flower Costume");
        bodyMap[46] = bytes32("Cold Chain");
        bodyMap[47] = bytes32("Straight Jacket");
        bodyMap[48] = bytes32("Predator Costume");
        bodyMap[49] = bytes32("White & Gold Dress");
        bodyMap[50] = bytes32("Spike Jacket");
        bodyMap[51] = bytes32("Invisibility Cloak");
        bodyMap[52] = bytes32("GOAT Jacket");
        bodyMap[53] = bytes32("Black & Blue Dress");

        // Initialize eyes map
        eyesMap[0] = bytes32("No traits");
        eyesMap[1] = bytes32("Clear Eyes");
        eyesMap[2] = bytes32("Sus");
        eyesMap[3] = bytes32("Scum");
        eyesMap[4] = bytes32("Shroomed");
        eyesMap[5] = bytes32("Shark");
        eyesMap[6] = bytes32("Cleo");
        eyesMap[7] = bytes32("Soft Glam");
        eyesMap[8] = bytes32("Goat");
        eyesMap[9] = bytes32("Cat");
        eyesMap[10] = bytes32("Clown");
        eyesMap[11] = bytes32("Tearful");
        eyesMap[12] = bytes32("Mesmerized");
        eyesMap[13] = bytes32("Scum Clown");
        eyesMap[14] = bytes32("Blind");
        eyesMap[15] = bytes32("Black Eye");
        eyesMap[16] = bytes32("Black Eye Scum");

        // Initialize head map
        headMap[0] = bytes32("No traits");
        headMap[1] = bytes32("Logo Cap");
        headMap[2] = bytes32("Beanie");
        headMap[3] = bytes32("Mullet");
        headMap[4] = bytes32("Dreadlocks");
        headMap[5] = bytes32("V-cut");
        headMap[6] = bytes32("Sleek");
        headMap[7] = bytes32("Trucker Hat");
        headMap[8] = bytes32("Preppy");
        headMap[9] = bytes32("Bun");
        headMap[10] = bytes32("Black Bucket Hat");
        headMap[11] = bytes32("Fade");
        headMap[12] = bytes32("Ushanka");
        headMap[13] = bytes32("Red Spikes");
        headMap[14] = bytes32("Beach");
        headMap[15] = bytes32("Afro");
        headMap[16] = bytes32("Lover Bug");
        headMap[17] = bytes32("New Jack City");
        headMap[18] = bytes32("Blume");
        headMap[19] = bytes32("Mohawk");
        headMap[20] = bytes32("Fitted Cap");
        headMap[21] = bytes32("Karen");
        headMap[22] = bytes32("Blond Bowlcut");
        headMap[23] = bytes32("The King");
        headMap[24] = bytes32("Durag");
        headMap[25] = bytes32("Bucket Hat");
        headMap[26] = bytes32("Bus Driver");
        headMap[27] = bytes32("Black Bowlcut");
        headMap[28] = bytes32("Rawr xd");
        headMap[29] = bytes32("Cowboy");
        headMap[30] = bytes32("Aerobics");
        headMap[31] = bytes32("Rattail");
        headMap[32] = bytes32("Clown");
        headMap[33] = bytes32("Hair Metal");
        headMap[34] = bytes32("Unibruh");
        headMap[35] = bytes32("Biker");
        headMap[36] = bytes32("Piercing");
        headMap[37] = bytes32("/");
        headMap[38] = bytes32("Knot Head");
        headMap[39] = bytes32("Say No More");
        headMap[40] = bytes32("Mixologist");
        headMap[41] = bytes32("Sahara Cap");
        headMap[42] = bytes32("Cheetah Fur Hat");
        headMap[43] = bytes32("Pharaoh");
        headMap[44] = bytes32("Giga Brain");
        headMap[45] = bytes32("Flower Costume");
        headMap[46] = bytes32("Rockstar");
        headMap[47] = bytes32("Dryden");
        headMap[48] = bytes32("Trash's King");
        headMap[49] = bytes32("Tin Foil Hat");
        headMap[50] = bytes32("Umbrella");
        headMap[51] = bytes32("Archive Cap");
        headMap[52] = bytes32("Rug Legend");
        headMap[53] = bytes32("Compost God");
        headMap[54] = bytes32("Balaclava");
        headMap[55] = bytes32("Predator Costume");

        // Initialize mouth map
        mouthMap[0] = bytes32("Smiley");
        mouthMap[1] = bytes32("Duck Face");
        mouthMap[2] = bytes32("Vortex");
        mouthMap[3] = bytes32("Pornstache");
        mouthMap[4] = bytes32("Ron");
        mouthMap[5] = bytes32("Bandana");
        mouthMap[6] = bytes32("Kiss");
        mouthMap[7] = bytes32("Glossy Lips");
        mouthMap[8] = bytes32("Hogan");
        mouthMap[9] = bytes32("Buckteeth");
        mouthMap[10] = bytes32("Lumbersexual");
        mouthMap[11] = bytes32("Grrr");
        mouthMap[12] = bytes32("Toothbrush");
        mouthMap[13] = bytes32("Rotten");
        mouthMap[14] = bytes32("Zoidbug");
        mouthMap[15] = bytes32("Reptilian");
        mouthMap[16] = bytes32("Goth");
        mouthMap[17] = bytes32("Tooth Cap");
        mouthMap[18] = bytes32("Lemmy");
        mouthMap[19] = bytes32("Gold Grillz");
        mouthMap[20] = bytes32("Overjet");
        mouthMap[21] = bytes32("Duke In Vegas");
        mouthMap[22] = bytes32("Hannibal");
        mouthMap[23] = bytes32("Iced Out");
        mouthMap[24] = bytes32("Braces");

        // Initialize background map
        backgroundMap[0] = bytes32("blue");
        backgroundMap[1] = bytes32("green");
        backgroundMap[2] = bytes32("orange");
        backgroundMap[3] = bytes32("pink");
        backgroundMap[4] = bytes32("red");
        backgroundMap[5] = bytes32("yellow");

        // Initialize bug type map
        bugTypeMap[0] = bytes32("mantis");
        bugTypeMap[1] = bytes32("caterpillar");
        bugTypeMap[2] = bytes32("fly");
        bugTypeMap[3] = bytes32("mosquito");
        bugTypeMap[4] = bytes32("moth");
        bugTypeMap[5] = bytes32("snail");

    }

}
// File: /contracts/ScumbugsMetadata.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "./ScumbugsValues.sol";
import "./DateTime.sol";

/**
 * @title Scumbugs Metadata storage
 * @notice Stores the metadata on-chain.
 */
contract ScumbugsMetadata is DateTime {

    // ---------------------- Structs ----------------------
    struct Attributes {
        uint8 hand;
        uint8 body;
        uint8 eyes;
        uint8 head;
        uint8 mouth;
        uint8 background_color;
        uint8 bug_type;
        uint32 birthday;
    }

    struct MetadataInput {
        uint256 tokenId;
        bytes32 txHash;
        bytes32 mediaId1;
        bytes32 mediaId2;
        bytes32 mediaBdayId1;
        bytes32 mediaBdayId2;
        // 11 bytes
        uint8 hand;
        uint8 body;
        uint8 eyes;
        uint8 head;
        uint8 mouth;
        uint8 background_color;
        uint8 bug_type;
        uint32 birthday;
    }

    mapping (uint256 => Attributes) internal attributesMap;
    mapping (uint256 => bytes32) public txhashes;
    mapping (uint256 => bytes32) internal mediaIds1;
    mapping (uint256 => bytes32) internal mediaIds2;
    mapping (uint256 => bytes32) internal mediaBdayIds1;
    mapping (uint256 => bytes32) internal mediaBdayIds2;
    bytes32 private defaultImagePart1 = bytes32("U95dUPeExh0CVYAH0ga8jOuifsPD6FsF");
    bytes32 private defaultImagePart2 = bytes32("6aGK2oGauo8");
    
    bytes32 internal siteUrl;
    address public ScumbugsValuesAddress;
    

    // ---------------------- Constructor ----------------------

    /**
     * @notice Constructor. Sets networkPath and siteUrl.
     * @param _siteUrl The value given to siteUrl
     * @param _ScumbugsValuesAddress The address of the Scumbug Values
     */
    constructor(bytes32 _siteUrl, address _ScumbugsValuesAddress) {
        siteUrl = _siteUrl;
        ScumbugsValuesAddress = _ScumbugsValuesAddress;
    }

    /**
     * @notice Renders json as a string containing metadata for token with tokenId
     */
    function _render(uint256 tokenId) internal view virtual returns (string memory) {
        bool isBirthday = false;
        uint8 nowDay = getDay(block.timestamp);
        uint8 nowMonth = getMonth(block.timestamp);
        uint8 bdDay = getDay(attributesMap[tokenId].birthday);
        uint8 bdMonth = getMonth(attributesMap[tokenId].birthday);
        if (nowDay == bdDay && nowMonth == bdMonth) {
            isBirthday = true;
        }
        delete nowDay;
        delete nowMonth;
        delete bdDay;
        delete bdMonth;
        bool generated = isGenerated(tokenId);
        bytes memory tokenIdBytes = uintToStrBytes(tokenId);
        bytes memory buffer = bytes.concat(bytes28("data:application/json;utf8,{"), bytes8('"name":"'));
        // Name
        buffer = bytes.concat(buffer, bytes9("Scumbug #"), tokenIdBytes, bytes2('",'));
        // Image
        if (!generated) {
            buffer = bytes.concat(buffer, bytes9('"image":"'), bytes5("ar://"), defaultImagePart1);
            buffer = bytes.concat(buffer, defaultImagePart2, bytes2('",'));
        } else if (isBirthday) {
            buffer = bytes.concat(buffer, bytes9('"image":"'), bytes5("ar://"), mediaBdayIds1[tokenId]);
            buffer = bytes.concat(buffer, mediaBdayIds2[tokenId], bytes2('",'));
        } else {
            buffer = bytes.concat(buffer, bytes9('"image":"'), bytes5("ar://"), mediaIds1[tokenId]);
            buffer = bytes.concat(buffer, mediaIds2[tokenId], bytes2('",'));
        }
        // Token Id
        buffer = bytes.concat(buffer, bytes10('"tokenId":'), tokenIdBytes, bytes1(","));
        if (generated) {
            // External URL
            buffer = bytes.concat(buffer, bytes16('"external_url":"'), siteUrl, tokenIdBytes, bytes2('",'));
            // Clean up
            delete tokenIdBytes;
            Attributes memory attributes = attributesMap[tokenId];
            ScumbugsValues scumbugsValuesInstance = ScumbugsValues(ScumbugsValuesAddress);
            // Open attributes
            buffer = bytes.concat(buffer, bytes14('"attributes":['));
            // hand
            if (attributes.hand != 0) {
                bytes32 data = scumbugsValuesInstance.handMap(attributes.hand);
                buffer = bytes.concat(buffer, bytes30('{"trait_type":"hand","value":"'), data, bytes3('"},'));
            }
            // body
            if (attributes.body != 0) {
                bytes32 data = scumbugsValuesInstance.bodyMap(attributes.body);
                buffer = bytes.concat(buffer, bytes30('{"trait_type":"body","value":"'), data, bytes3('"},'));
            }
            // eyes
            if (attributes.eyes != 0) {
                bytes32 data = scumbugsValuesInstance.eyesMap(attributes.eyes);
                buffer = bytes.concat(buffer, bytes30('{"trait_type":"eyes","value":"'), data, bytes3('"},'));
            }
            // head
            if (attributes.head != 0) {
                bytes32 data = scumbugsValuesInstance.headMap(attributes.head);
                buffer = bytes.concat(buffer, bytes30('{"trait_type":"head","value":"'), data, bytes3('"},'));
            }
            // mouth
            if (attributes.mouth != 0) {
                bytes32 data = scumbugsValuesInstance.mouthMap(attributes.mouth);
                buffer = bytes.concat(buffer, bytes31('{"trait_type":"mouth","value":"'), data, bytes3('"},'));
            }
            // Background color
            if (isBirthday) {
                buffer = bytes.concat(buffer, bytes32('{"trait_type":"background_color"'), bytes22(',"value":"Birthday!"},'));
            } else {
                bytes32 backgroundData = scumbugsValuesInstance.backgroundMap(attributes.background_color);
                buffer = bytes.concat(buffer, bytes32('{"trait_type":"background_color"'), bytes10(',"value":"'), backgroundData, bytes3('"},'));
                delete backgroundData;
            }
            // Bug type
            bytes32 bugTypeData = scumbugsValuesInstance.bugTypeMap(attributes.bug_type);
            buffer = bytes.concat(buffer, bytes24('{"trait_type":"bug_type"'), bytes10(',"value":"'), bugTypeData, bytes3('"},'));
            delete bugTypeData;
            delete scumbugsValuesInstance;
            // Birthday
            buffer = bytes.concat(buffer, bytes25('{"trait_type":"birthday",'), bytes22('"display_type":"date",'), bytes8('"value":'), 
                uintToStrBytes(attributes.birthday), bytes1('}'));
            // Close attributes
            delete attributes;
            buffer = bytes.concat(buffer, bytes1("]"));
        } else {
            buffer = bytes.concat(buffer, bytes15('"attributes":[]'));
        }
        // Close json
        buffer = bytes.concat(buffer, bytes1("}"));
        return string(buffer);
    }

    function isGenerated(uint256 tokenId) public view returns (bool) {
        return mediaIds1[tokenId] != 0;
    }

    /**
     * @notice Transforms uint to string bytes. Based on toString(uint256 value) from @openzeppelin/contracts/utils/Strings.sol
     */
    function uintToStrBytes(uint256 value) pure private returns (bytes memory) {
        if (value == 0) {
            return bytes("0");
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return buffer;
    }

}
// File: /contracts/Pausable.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";

/**
 * @title Pausable
 * @dev Base contract which allows children to implement an emergency stop mechanism.
 */
contract Pausable is Ownable {

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }


    /**
    * @dev Modifier to make a function callable only when the contract is not paused.
    */
    modifier whenNotPaused() {
        require(!_paused);
        _;
    }

    /**
    * @dev Modifier to make a function callable only when the contract is paused.
    */
    modifier whenPaused() {
        require(_paused);
        _;
    }

    /**
    * @dev called by the owner to pause, triggers stopped state
    */
    function pause() onlyOwner whenNotPaused public {
        _paused = true;
    }

    /**
    * @dev called by the owner to unpause, returns to normal state
    */
    function unpause() onlyOwner whenPaused public {
        _paused = false;
    }
}
// File: /contracts/OpenseaProxy.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

// These contract definitions are used to create a reference to the OpenSea
// ProxyRegistry contract by using the registry's address (see isApprovedForAll).
interface OwnableDelegateProxy {

}

interface ProxyRegistry {
    function proxies(address) external view returns (OwnableDelegateProxy);
}
// File: /contracts/DateTime.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

/**
* Inspired by DateTime: https://etherscan.io/address/0x1a6184cd4c5bea62b0116de7962ee7315b7bcbce#code
*/
contract DateTime {
        /*
         *  Date and Time utilities for ethereum contracts
         *
         */
        struct _DateTime {
                uint16 year;
                uint8 month;
                uint8 day;
        }

        uint private constant DAY_IN_SECONDS = 86400;
        uint private constant YEAR_IN_SECONDS = 31536000;
        uint private constant LEAP_YEAR_IN_SECONDS = 31622400;

        uint private constant HOUR_IN_SECONDS = 3600;
        uint private constant MINUTE_IN_SECONDS = 60;

        uint16 private constant ORIGIN_YEAR = 1970;

        function isLeapYear(uint16 year) internal pure returns (bool) {
                if (year % 4 != 0) {
                        return false;
                }
                if (year % 100 != 0) {
                        return true;
                }
                if (year % 400 != 0) {
                        return false;
                }
                return true;
        }

        function leapYearsBefore(uint year) internal pure returns (uint) {
                year -= 1;
                return year / 4 - year / 100 + year / 400;
        }

        function getDaysInMonth(uint8 month, uint16 year) internal pure returns (uint8) {
                if (month == 1 || month == 3 || month == 5 || month == 7 || month == 8 || month == 10 || month == 12) {
                        return 31;
                }
                else if (month == 4 || month == 6 || month == 9 || month == 11) {
                        return 30;
                }
                else if (isLeapYear(year)) {
                        return 29;
                }
                else {
                        return 28;
                }
        }

        function parseTimestamp(uint timestamp) internal pure returns (_DateTime memory dt) {
                uint secondsAccountedFor = 0;
                uint buf;
                uint8 i;

                // Year
                dt.year = getYear(timestamp);
                buf = leapYearsBefore(dt.year) - leapYearsBefore(ORIGIN_YEAR);

                secondsAccountedFor += LEAP_YEAR_IN_SECONDS * buf;
                secondsAccountedFor += YEAR_IN_SECONDS * (dt.year - ORIGIN_YEAR - buf);

                // Month
                uint secondsInMonth;
                for (i = 1; i <= 12; i++) {
                        secondsInMonth = DAY_IN_SECONDS * getDaysInMonth(i, dt.year);
                        if (secondsInMonth + secondsAccountedFor > timestamp) {
                                dt.month = i;
                                break;
                        }
                        secondsAccountedFor += secondsInMonth;
                }

                // Day
                for (i = 1; i <= getDaysInMonth(dt.month, dt.year); i++) {
                        if (DAY_IN_SECONDS + secondsAccountedFor > timestamp) {
                                dt.day = i;
                                break;
                        }
                        secondsAccountedFor += DAY_IN_SECONDS;
                }
        }

        function getYear(uint timestamp) internal pure returns (uint16) {
                uint secondsAccountedFor = 0;
                uint16 year;
                uint numLeapYears;

                // Year
                year = uint16(ORIGIN_YEAR + timestamp / YEAR_IN_SECONDS);
                numLeapYears = leapYearsBefore(year) - leapYearsBefore(ORIGIN_YEAR);

                secondsAccountedFor += LEAP_YEAR_IN_SECONDS * numLeapYears;
                secondsAccountedFor += YEAR_IN_SECONDS * (year - ORIGIN_YEAR - numLeapYears);

                while (secondsAccountedFor > timestamp) {
                        if (isLeapYear(uint16(year - 1))) {
                                secondsAccountedFor -= LEAP_YEAR_IN_SECONDS;
                        }
                        else {
                                secondsAccountedFor -= YEAR_IN_SECONDS;
                        }
                        year -= 1;
                }
                return year;
        }

        function getMonth(uint timestamp) internal pure returns (uint8) {
                return parseTimestamp(timestamp).month;
        }

        function getDay(uint timestamp) internal pure returns (uint8) {
                return parseTimestamp(timestamp).day;
        }
}
// File: erc721a/contracts/ERC721A.sol



// SPDX-License-Identifier: MIT
// Creator: Chiru Labs

pragma solidity ^0.8.4;

import '@openzeppelin/contracts/token/ERC721/IERC721.sol';
import '@openzeppelin/contracts/token/ERC721/IERC721Receiver.sol';
import '@openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol';
import '@openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol';
import '@openzeppelin/contracts/utils/Address.sol';
import '@openzeppelin/contracts/utils/Context.sol';
import '@openzeppelin/contracts/utils/Strings.sol';
import '@openzeppelin/contracts/utils/introspection/ERC165.sol';

error ApprovalCallerNotOwnerNorApproved();
error ApprovalQueryForNonexistentToken();
error ApproveToCaller();
error ApprovalToCurrentOwner();
error BalanceQueryForZeroAddress();
error MintedQueryForZeroAddress();
error BurnedQueryForZeroAddress();
error MintToZeroAddress();
error MintZeroQuantity();
error OwnerIndexOutOfBounds();
error OwnerQueryForNonexistentToken();
error TokenIndexOutOfBounds();
error TransferCallerNotOwnerNorApproved();
error TransferFromIncorrectOwner();
error TransferToNonERC721ReceiverImplementer();
error TransferToZeroAddress();
error URIQueryForNonexistentToken();

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata and Enumerable extension. Built to optimize for lower gas during batch mints.
 *
 * Assumes serials are sequentially minted starting at 0 (e.g. 0, 1, 2, 3..).
 *
 * Assumes that an owner cannot have more than 2**64 - 1 (max value of uint64) of supply.
 *
 * Assumes that the maximum token id cannot exceed 2**128 - 1 (max value of uint128).
 */
contract ERC721A is Context, ERC165, IERC721, IERC721Metadata, IERC721Enumerable {
    using Address for address;
    using Strings for uint256;

    // Compiler will pack this into a single 256bit word.
    struct TokenOwnership {
        // The address of the owner.
        address addr;
        // Keeps track of the start time of ownership with minimal overhead for tokenomics.
        uint64 startTimestamp;
        // Whether the token has been burned.
        bool burned;
    }

    // Compiler will pack this into a single 256bit word.
    struct AddressData {
        // Realistically, 2**64-1 is more than enough.
        uint64 balance;
        // Keeps track of mint count with minimal overhead for tokenomics.
        uint64 numberMinted;
        // Keeps track of burn count with minimal overhead for tokenomics.
        uint64 numberBurned;
    }

    // Compiler will pack the following 
    // _currentIndex and _burnCounter into a single 256bit word.
    
    // The tokenId of the next token to be minted.
    uint128 internal _currentIndex;

    // The number of tokens burned.
    uint128 internal _burnCounter;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to ownership details
    // An empty struct value does not necessarily mean the token is unowned. See ownershipOf implementation for details.
    mapping(uint256 => TokenOwnership) internal _ownerships;

    // Mapping owner address to address data
    mapping(address => AddressData) private _addressData;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC721Enumerable-totalSupply}.
     */
    function totalSupply() public view override returns (uint256) {
        // Counter underflow is impossible as _burnCounter cannot be incremented
        // more than _currentIndex times
        unchecked {
            return _currentIndex - _burnCounter;    
        }
    }

    /**
     * @dev See {IERC721Enumerable-tokenByIndex}.
     * This read function is O(totalSupply). If calling from a separate contract, be sure to test gas first.
     * It may also degrade with extremely large collection sizes (e.g >> 10000), test for your use case.
     */
    function tokenByIndex(uint256 index) public view override returns (uint256) {
        uint256 numMintedSoFar = _currentIndex;
        uint256 tokenIdsIdx;

        // Counter overflow is impossible as the loop breaks when
        // uint256 i is equal to another uint256 numMintedSoFar.
        unchecked {
            for (uint256 i; i < numMintedSoFar; i++) {
                TokenOwnership memory ownership = _ownerships[i];
                if (!ownership.burned) {
                    if (tokenIdsIdx == index) {
                        return i;
                    }
                    tokenIdsIdx++;
                }
            }
        }
        revert TokenIndexOutOfBounds();
    }

    /**
     * @dev See {IERC721Enumerable-tokenOfOwnerByIndex}.
     * This read function is O(totalSupply). If calling from a separate contract, be sure to test gas first.
     * It may also degrade with extremely large collection sizes (e.g >> 10000), test for your use case.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) public view override returns (uint256) {
        if (index >= balanceOf(owner)) revert OwnerIndexOutOfBounds();
        uint256 numMintedSoFar = _currentIndex;
        uint256 tokenIdsIdx;
        address currOwnershipAddr;

        // Counter overflow is impossible as the loop breaks when
        // uint256 i is equal to another uint256 numMintedSoFar.
        unchecked {
            for (uint256 i; i < numMintedSoFar; i++) {
                TokenOwnership memory ownership = _ownerships[i];
                if (ownership.burned) {
                    continue;
                }
                if (ownership.addr != address(0)) {
                    currOwnershipAddr = ownership.addr;
                }
                if (currOwnershipAddr == owner) {
                    if (tokenIdsIdx == index) {
                        return i;
                    }
                    tokenIdsIdx++;
                }
            }
        }

        // Execution should never reach this point.
        revert();
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            interfaceId == type(IERC721Enumerable).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view override returns (uint256) {
        if (owner == address(0)) revert BalanceQueryForZeroAddress();
        return uint256(_addressData[owner].balance);
    }

    function _numberMinted(address owner) internal view returns (uint256) {
        if (owner == address(0)) revert MintedQueryForZeroAddress();
        return uint256(_addressData[owner].numberMinted);
    }

    function _numberBurned(address owner) internal view returns (uint256) {
        if (owner == address(0)) revert BurnedQueryForZeroAddress();
        return uint256(_addressData[owner].numberBurned);
    }

    /**
     * Gas spent here starts off proportional to the maximum mint batch size.
     * It gradually moves to O(1) as tokens get transferred around in the collection over time.
     */
    function ownershipOf(uint256 tokenId) internal view returns (TokenOwnership memory) {
        uint256 curr = tokenId;

        unchecked {
            if (curr < _currentIndex) {
                TokenOwnership memory ownership = _ownerships[curr];
                if (!ownership.burned) {
                    if (ownership.addr != address(0)) {
                        return ownership;
                    }
                    // Invariant: 
                    // There will always be an ownership that has an address and is not burned 
                    // before an ownership that does not have an address and is not burned.
                    // Hence, curr will not underflow.
                    while (true) {
                        curr--;
                        ownership = _ownerships[curr];
                        if (ownership.addr != address(0)) {
                            return ownership;
                        }
                    }
                }
            }
        }
        revert OwnerQueryForNonexistentToken();
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view override returns (address) {
        return ownershipOf(tokenId).addr;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        if (!_exists(tokenId)) revert URIQueryForNonexistentToken();

        string memory baseURI = _baseURI();
        return bytes(baseURI).length != 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : '';
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return '';
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public override {
        address owner = ERC721A.ownerOf(tokenId);
        if (to == owner) revert ApprovalToCurrentOwner();

        if (_msgSender() != owner && !isApprovedForAll(owner, _msgSender())) {
            revert ApprovalCallerNotOwnerNorApproved();
        }

        _approve(to, tokenId, owner);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view override returns (address) {
        if (!_exists(tokenId)) revert ApprovalQueryForNonexistentToken();

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public override {
        if (operator == _msgSender()) revert ApproveToCaller();

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, '');
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        _transfer(from, to, tokenId);
        if (!_checkOnERC721Received(from, to, tokenId, _data)) {
            revert TransferToNonERC721ReceiverImplementer();
        }
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     */
    function _exists(uint256 tokenId) internal view returns (bool) {
        return tokenId < _currentIndex && !_ownerships[tokenId].burned;
    }

    function _safeMint(address to, uint256 quantity) internal {
        _safeMint(to, quantity, '');
    }

    /**
     * @dev Safely mints `quantity` tokens and transfers them to `to`.
     *
     * Requirements:
     *
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called for each safe transfer.
     * - `quantity` must be greater than 0.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(
        address to,
        uint256 quantity,
        bytes memory _data
    ) internal {
        _mint(to, quantity, _data, true);
    }

    /**
     * @dev Mints `quantity` tokens and transfers them to `to`.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `quantity` must be greater than 0.
     *
     * Emits a {Transfer} event.
     */
    function _mint(
        address to,
        uint256 quantity,
        bytes memory _data,
        bool safe
    ) internal {
        uint256 startTokenId = _currentIndex;
        if (to == address(0)) revert MintToZeroAddress();
        if (quantity == 0) revert MintZeroQuantity();

        _beforeTokenTransfers(address(0), to, startTokenId, quantity);

        // Overflows are incredibly unrealistic.
        // balance or numberMinted overflow if current value of either + quantity > 3.4e38 (2**128) - 1
        // updatedIndex overflows if _currentIndex + quantity > 3.4e38 (2**128) - 1
        unchecked {
            _addressData[to].balance += uint64(quantity);
            _addressData[to].numberMinted += uint64(quantity);

            _ownerships[startTokenId].addr = to;
            _ownerships[startTokenId].startTimestamp = uint64(block.timestamp);

            uint256 updatedIndex = startTokenId;

            for (uint256 i; i < quantity; i++) {
                emit Transfer(address(0), to, updatedIndex);
                if (safe && !_checkOnERC721Received(address(0), to, updatedIndex, _data)) {
                    revert TransferToNonERC721ReceiverImplementer();
                }
                updatedIndex++;
            }

            _currentIndex = uint128(updatedIndex);
        }
        _afterTokenTransfers(address(0), to, startTokenId, quantity);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) private {
        TokenOwnership memory prevOwnership = ownershipOf(tokenId);

        bool isApprovedOrOwner = (_msgSender() == prevOwnership.addr ||
            isApprovedForAll(prevOwnership.addr, _msgSender()) ||
            getApproved(tokenId) == _msgSender());

        if (!isApprovedOrOwner) revert TransferCallerNotOwnerNorApproved();
        if (prevOwnership.addr != from) revert TransferFromIncorrectOwner();
        if (to == address(0)) revert TransferToZeroAddress();

        _beforeTokenTransfers(from, to, tokenId, 1);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId, prevOwnership.addr);

        // Underflow of the sender's balance is impossible because we check for
        // ownership above and the recipient's balance can't realistically overflow.
        // Counter overflow is incredibly unrealistic as tokenId would have to be 2**128.
        unchecked {
            _addressData[from].balance -= 1;
            _addressData[to].balance += 1;

            _ownerships[tokenId].addr = to;
            _ownerships[tokenId].startTimestamp = uint64(block.timestamp);

            // If the ownership slot of tokenId+1 is not explicitly set, that means the transfer initiator owns it.
            // Set the slot of tokenId+1 explicitly in storage to maintain correctness for ownerOf(tokenId+1) calls.
            uint256 nextTokenId = tokenId + 1;
            if (_ownerships[nextTokenId].addr == address(0)) {
                // This will suffice for checking _exists(nextTokenId),
                // as a burned slot cannot contain the zero address.
                if (nextTokenId < _currentIndex) {
                    _ownerships[nextTokenId].addr = prevOwnership.addr;
                    _ownerships[nextTokenId].startTimestamp = prevOwnership.startTimestamp;
                }
            }
        }

        emit Transfer(from, to, tokenId);
        _afterTokenTransfers(from, to, tokenId, 1);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        TokenOwnership memory prevOwnership = ownershipOf(tokenId);

        _beforeTokenTransfers(prevOwnership.addr, address(0), tokenId, 1);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId, prevOwnership.addr);

        // Underflow of the sender's balance is impossible because we check for
        // ownership above and the recipient's balance can't realistically overflow.
        // Counter overflow is incredibly unrealistic as tokenId would have to be 2**128.
        unchecked {
            _addressData[prevOwnership.addr].balance -= 1;
            _addressData[prevOwnership.addr].numberBurned += 1;

            // Keep track of who burned the token, and the timestamp of burning.
            _ownerships[tokenId].addr = prevOwnership.addr;
            _ownerships[tokenId].startTimestamp = uint64(block.timestamp);
            _ownerships[tokenId].burned = true;

            // If the ownership slot of tokenId+1 is not explicitly set, that means the burn initiator owns it.
            // Set the slot of tokenId+1 explicitly in storage to maintain correctness for ownerOf(tokenId+1) calls.
            uint256 nextTokenId = tokenId + 1;
            if (_ownerships[nextTokenId].addr == address(0)) {
                // This will suffice for checking _exists(nextTokenId),
                // as a burned slot cannot contain the zero address.
                if (nextTokenId < _currentIndex) {
                    _ownerships[nextTokenId].addr = prevOwnership.addr;
                    _ownerships[nextTokenId].startTimestamp = prevOwnership.startTimestamp;
                }
            }
        }

        emit Transfer(prevOwnership.addr, address(0), tokenId);
        _afterTokenTransfers(prevOwnership.addr, address(0), tokenId, 1);

        // Overflow not possible, as _burnCounter cannot be exceed _currentIndex times.
        unchecked { 
            _burnCounter++;
        }
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(
        address to,
        uint256 tokenId,
        address owner
    ) private {
        _tokenApprovals[tokenId] = to;
        emit Approval(owner, to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver(to).onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert TransferToNonERC721ReceiverImplementer();
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before a set of serially-ordered token ids are about to be transferred. This includes minting.
     * And also called before burning one token.
     *
     * startTokenId - the first token id to be transferred
     * quantity - the amount to be transferred
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, `from`'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, `tokenId` will be burned by `from`.
     * - `from` and `to` are never both zero.
     */
    function _beforeTokenTransfers(
        address from,
        address to,
        uint256 startTokenId,
        uint256 quantity
    ) internal virtual {}

    /**
     * @dev Hook that is called after a set of serially-ordered token ids have been transferred. This includes
     * minting.
     * And also called after one token has been burned.
     *
     * startTokenId - the first token id to be transferred
     * quantity - the amount to be transferred
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, `from`'s `tokenId` has been
     * transferred to `to`.
     * - When `from` is zero, `tokenId` has been minted for `to`.
     * - When `to` is zero, `tokenId` has been burned by `from`.
     * - `from` and `to` are never both zero.
     */
    function _afterTokenTransfers(
        address from,
        address to,
        uint256 startTokenId,
        uint256 quantity
    ) internal virtual {}
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Enumerable.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional enumeration extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Enumerable is IERC721 {
    /**
     * @dev Returns the total amount of tokens stored by the contract.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns a token ID owned by `owner` at a given `index` of its token list.
     * Use along with {balanceOf} to enumerate all of ``owner``'s tokens.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) external view returns (uint256 tokenId);

    /**
     * @dev Returns a token ID at a given `index` of all the tokens stored by the contract.
     * Use along with {totalSupply} to enumerate all tokens.
     */
    function tokenByIndex(uint256 index) external view returns (uint256);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}